Компьютерная математика

УДК 519.85

П.І. СТЕЦЮК, О.П. БИСАГА, С.С. ТРЕГУБЕНКО

ДВОЕТАПНА ТРАНСПОРТНА ЗАДАЧА З ОБМЕЖЕННЯМ НА КІЛЬКІСТЬ ПРОМІЖНИХ ПУНКТІВ

Запропонована математична модель двоетапної транспортної задачі для визначення найекономічнішого плану перевезення однорідної продукції від постачальників до споживачів, якщо кількість проміжних пунктів є обмеженою зверху. Математична модель сформульована як задача булевого лінійного програмування. Визначено умови, при яких задача має розв'язок, та наведено AMPL-код для її розв'язання сучасними солверами лінійного цілочислового програмування. Наведено демонстраційний приклад з результатами розрахунку за допомогою програми gurobi.

Вступ. Математичні моделі задач виробничо-транспортного типу відіграють значну роль при оптимальному плануванні різного роду процесів транспортування та зберігання виробленої продукції або будь-яких вантажів та предметів [1]. Ці задачі мають ряд властивостей, близьких до транспортної задачі, що дозволяє розробляти спеціальні методи їх розв'язання, які зазвичай більш ефективні, ніж загальні методи лінійного програмування. Властивості виробничо-транспортних задач використовуються в схемах декомпозиції, які в поєднанні з методами негладкої оптимізації дозволяють отримати максимальний обчислювальний ефект [1]. Це стосується і двоетапних транспортних задач, в яких вантаж перевозиться від постачальників до споживачів тільки через проміжні пункти.

У даній статті розглядається та досліджується модифікація двоетапної транспортної задачі для знаходження найекономічнішого плану перевезення однорідної продукції від постачальників до споживачів за умови, що кількість проміжних пунктів є обмеженою зверху. Ця модифікація – актуальна для агропідприємств при розподіленні та доставці вирощеної продукції для продажу або переробки на власних потужностях. Як проміжні пункти тут виступають власні та орендовані елеватори (зерносховища). Вона може знайти застосування для пошуку раціонального розташування заданої кількості складів з урахуванням визначеного положення постачальників та отримувачів матеріально-технічних засобів на території, де вони виконують свої завдання.

1. Формулювання задачі. Нехай в m пунктах постачання $A_1,...,A_m$ є $a_1,...,a_m$ одиниць продукції, яку потрібно перевезти до n споживачів $B_1,...,B_n$, задовольнивши їх потреби $b_1,...,b_n$. Для транспортування продукції від постачальників до споживачів можна задіяти l проміжних пунктів $D_1,...,D_l$, пропускні спроможності яких будемо вважати необмеженими. Потрібно знайти оптимальний план транспортування продукції, який використовує не більше, ніж d ($1 \le d \le l$) проміжних пунктів, де c_{ik} – витрати на перевезення одиниці продукції від постачальника A_i до проміжного пункту D_k, а c_{kj} – витрати на перевезення одиниці продукції від проміжного пункту D_k до споживача B_j.

Нехай $x = \{x_{ik}\}_{i=1,\dots,m}^{k=1,\dots,l}$, де x_{ik} – кількість одиниць продукції, яка перевозиться від постачальника A_i до пункту D_k; $y = \{y_{kj}\}_{k=1,\dots,l}^{j=1,\dots,n}$, де y_{kj} – кількість продукції від пункту D_k до споживача B_j; $z = \{z_k\}_{k=1,\dots,l}$, де z_k – булева змінна, яка дорівнює одиниці, якщо проміжний пункт D_k використовується, та дорівнює нулю – в протилежному випадку.

Двоетапна транспортна задача для знаходження не більше, ніж d проміжних пунктів, які визначають найекономічніший план перевезення продукції від постачальників до споживачів, має такий вигляд: знайти

$$f_{xyz}^* = f_{xyz}(x^*, y^*, z^*) = \min_{x,y,z} \left\{ f(x,y) = \sum_{i=1}^{m} \sum_{k=1}^{l} c_{ik} x_{ik} + \sum_{k=1}^{l} \sum_{j=1}^{n} c_{kj} y_{kj} \right\} \qquad (1)$$

за наступних обмежень

$$\sum_{k=1}^{l} x_{ik} = a_i, \quad i = 1,\dots,m, \qquad (2)$$

$$\sum_{k=1}^{l} y_{kj} = b_j, \quad j = 1,\dots,n, \qquad (3)$$

$$\sum_{i=1}^{m} x_{ik} - \sum_{j=1}^{n} y_{kj} = 0, \quad k = 1,\dots,l, \qquad (4)$$

$$\sum_{i=1}^{m} x_{ik} \le z_k \sum_{i=1}^{m} a_i, \quad k = 1,\dots,l, \qquad (5)$$

$$\sum_{k=1}^{l} z_k \le d, \qquad (6)$$

$$x_{ik} \ge 0,\ y_{kj} \ge 0,\ z_k = 0 \vee 1,\ i = 1,\dots,m,\ k = 1,\dots,l,\ j = 1,\dots,n. \qquad (7)$$

Задача (1) – (7) – задача булевого лінійного програмування, яка містить $(m+n)\times l$ неперервних змінних x та y, l булевих змінних z, $m+n+2l+1$ обмежень. Цільова функція (1) задає сумарні витрати на транспортування продукції від постачальників до споживачів. Обмеження (2) означають транспортування $a_1,...,a_m$ одиниць продукції із пунктів постачання до проміжних пунктів, а обмеження (3) – що споживачам потрібно доставити необхідні обсяги $b_1,...,b_n$ одиниць продукції з проміжних пунктів. Обмеження (4) задають умови на те, щоб вся продукція, яка приходить від постачальників до кожного проміжного пункту, була обов'язково відправлена споживачам. Обмеження (6) означає, що задіяними мають бути не більше, ніж d проміжних пунктів, а обмеження (5) визначають, які саме проміжні пункти будуть задіяні.

Якщо для задачі (1) – (7) вибрати $d = l$, що означає, що при транспортуванні продукції задіяними можуть бути усі l проміжних пунктів, то тоді обмеження (5) та (6) можна прибрати. В результаті отримаємо добре вивчену класичну двоетапну транспортну задачу [2, 3]. Тому справедливі наступні твердження.

Теорема 1. Якщо $d = l$, то задача $(1)-(7)$ переходить в двоетапну транспортну задачу: знайти

$$f_{xy}^* = f_{xy}(x^*, y^*) = \min_{x,y}\left\{ f(x,y) = \sum_{i=1}^{m}\sum_{k=1}^{l} c_{ik}x_{ik} + \sum_{k=1}^{l}\sum_{j=1}^{n} c_{kj}y_{kj} \right\} \tag{8}$$

за обмежень

$$\sum_{k=1}^{l} x_{ik} = a_i, \quad i = 1,\ldots,m, \tag{9}$$

$$\sum_{k=1}^{l} y_{kj} = b_j, \quad j = 1,\ldots,n, \tag{10}$$

$$\sum_{i=1}^{m} x_{ik} - \sum_{j=1}^{n} y_{kj} = 0, \quad k = 1,\ldots,l, \tag{11}$$

$$x_{ik} \ge 0,\ y_{kj} \ge 0, \quad i = 1,\ldots,m,\ k = 1,\ldots,l,\ j = 1,\ldots,n. \tag{12}$$

Теорема 2 [4]. Система обмежень (9) – (12) – несумісна, якщо $\sum_{i=1}^{m} a_i \ne \sum_{j=1}^{n} b_j$.

Якщо в задачі (8) – (12) загальна кількість продукції постачальників дорівнює загальному попиту всіх споживачів, тобто виконується рівність

$$\sum_{i=1}^{m} a_i = \sum_{j=1}^{n} b_j, \tag{13}$$

то таку двоетапну транспортну задачу називають збалансованою, або закритою. Якщо умова (13) не виконується, то двоетапну транспортну задачу називають незбалансованою або відкритою. Останню легко звести до задачі закритого типу: у разі перевищення загального попиту над запасами – за допомогою введення фіктивного постачальника A_{m+1} із кількістю $a_{m+1} = \sum_{j=1}^{n} b_j - \sum_{i=1}^{m} a_i$ одиниць продукції; у разі перевищення запасів над загальним попитом – за допомогою введення фіктивного споживача B_{n+1} з потребою $b_{n+1} = \sum_{i=1}^{m} a_i - \sum_{j=1}^{n} b_j$ одиниць продукції.

2. Про властивості задачі (1) – (7). Згідно теореми 1 задача (1) – (7) співпадає з класичною двоетапною транспортною задачею, якщо $d = l$, $1 \le d \le l$. Цей випадок відповідає тому, що d вибирається рівним правій границі інтервалу $[1, l]$ та означає, що при транспортуванні продукції від постачальників до споживачів задіяними можуть бути усі l проміжних пунктів. Якщо $d = 1$, то d вибирається рівним лівій границі інтервалу $[1, l]$, та означає, що всі постачальники відправляють продукцію у єдиний проміжний пункт, з якого кожний споживач забирає свою кількість одиниць продукції. Якщо задача (1) – (7) збалансована, тобто виконується умова (13), то граничному випадку $d = 1$ відповідає єдиний оптимальний план $x^* = \{x_{i1}^* = a_i\}_{i=1,\dots,m}$, $y^* = \{y_{1j}^* = b_j\}_{j=1,\dots,n}$, при якому реалізується мінімальне значення цільової функції $f_{xy}^* = \sum_{i=1}^{m} c_{i1} a_i + \sum_{j=1}^{n} c_{1j} b_j$.

Якщо виконується умова $1 < d < l$, то тоді задача (1) – (7) буде задачею лінійного булевого програмування. В обох крайніх випадках задача булевого лінійного програмування фактично відсутня, так як значення булевих змінних z визначаються однозначно. У першому випадку всі невідомі компоненти z дорівнюють одиниці, а у другому випадку – єдина булева змінна приймає значення, яке дорівнює одиниці.

Теорема 3. Система обмежень (2) – (7) є несумісною, якщо $\sum_{i=1}^{m} a_i \ne \sum_{j=1}^{n} b_j$.

Доведення. Доведення проведемо методом від супротивного. Якщо d таке, що $1 < d < l$, то виберемо довільні d значень булевих змінних z рівними одиниці, а інші $l - d$ булевих змінних – рівними нулю. Такий їх вибір задовольняє обмеженню (6). Для нульових булевих змінних (незадіяні проміжні пункти) обмеження (5) визначає нульові обсяги продукції від кожного постачальника до відповідного проміжного пункту та від відповідного проміжного пункту до кожного споживача. Тому для вибраних d проміжних пунктів система обмежень (2) – (7) переходить у наступну систему лінійних рівностей та нерівностей

$$\sum_{k=1}^{d} x_{ik} = a_i, \quad i = 1, \dots, m, \tag{14}$$

$$\sum_{k=1}^{d} y_{kj} = b_j, \quad j = 1, \dots, n, \tag{15}$$

$$\sum_{i=1}^{m} x_{ik} - \sum_{j=1}^{n} y_{kj} = 0, \quad k = 1, \dots, d, \tag{16}$$

$$x_{ik} \ge 0,\ y_{kj} \ge 0, \quad i = 1, \dots, m,\ k = 1, \dots, d,\ j = 1, \dots, n. \tag{17}$$

Приймемо, що існують невід’ємні $\overline{x} = \{\overline{x}_{ik}, i = 1,\ldots,m, k = 1,\ldots,d\}$ та $\overline{y} = \{\overline{y}_{kj}, k = 1,\ldots,d, j = 1,\ldots,n\}$, які задовольняють системі обмежень (14) – (17). Тобто для $\overline{x} \geq 0$ та $\overline{y} \geq 0$ справедливими є такі рівності:

$$\sum_{k=1}^{d} \overline{x}_{ik} = a_i, \quad i = 1,\ldots,m, \tag{18}$$

$$\sum_{k=1}^{d} \overline{y}_{kj} = b_j, \quad j = 1,\ldots,n, \tag{19}$$

$$\sum_{i=1}^{m} \overline{x}_{ik} - \sum_{j=1}^{n} \overline{y}_{kj} = 0, \quad k = 1,\ldots,d. \tag{20}$$

Якщо всі рівності із (18) просумувати за індексом $i = 1,\ldots,m,$ а всі рівності з (19) просумувати за індексом $j = 1,\ldots,n,$ то отримаємо такі рівності

$$\sum_{i=1}^{m}\sum_{k=1}^{d} \overline{x}_{ik} = \sum_{i=1}^{m} a_i, \tag{21}$$

$$\sum_{j=1}^{n}\sum_{k=1}^{d} \overline{y}_{jk} = \sum_{j=1}^{n} b_j. \tag{22}$$

Віднявши від рівності (21) рівність (22) та змінивши порядок індексів у сумах, отримаємо наступну рівність:

$$\sum_{k=1}^{d}\sum_{i=1}^{m} \overline{x}_{ik} - \sum_{k=1}^{d}\sum_{j=1}^{n} \overline{y}_{jk} = \sum_{i=1}^{m} a_i - \sum_{j=1}^{n} b_j. \tag{23}$$

Якщо всі рівності із (20) просумувати за індексом $k = 1,\ldots,d$, то отримаємо таку рівність:

$$\sum_{k=1}^{d}\sum_{i=1}^{m} \overline{x}_{ik} - \sum_{k=1}^{d}\sum_{j=1}^{n} \overline{y}_{kj} = 0. \tag{24}$$

За умови $\sum_{i=1}^{m} a_i \neq \sum_{j=1}^{n} b_j$ із рівності (23) випливає $\sum_{k=1}^{d}\sum_{i=1}^{m} \overline{x}_{ik} - \sum_{k=1}^{d}\sum_{j=1}^{n} \overline{y}_{kj} \neq 0$, що протирічить рівності (24). Це протиріччя доводить, що вектори $\overline{x}$ та $\overline{y}$ не можуть бути допустимими для обмежень (14) – (17), а значить при виконанні умови $\sum_{i=1}^{m} a_i \neq \sum_{j=1}^{n} b_j$ система обмежень (14) – (17) – несумісна. Враховуючи, що остання має місце для довільних d одиничних булевих змінних, то умова $\sum_{i=1}^{m} a_i \neq \sum_{j=1}^{n} b_j$ гарантує те, що система обмежень (2) – (7) – несумісна. Теорема доведена.

Теорема 3 дає можливість перевірити коректність вхідних даних для задачі (1) – (7). Іншими словами, вона дає можливість впевнитися, що задача (1) – (7) – збалансована. Якщо задача збалансована, то система (2) – (7) – сумісна, і переходимо до розв'язання задачі (1) – (7). Якщо задача – незбалансована, то закінчуємо розрахунок з повідомленням, що «задача (1) – (7) немає розв'язку».

3. AMPL-реалізація задачі (1) – (7). Один із можливих способів розв'язання задачі (1) – (7) полягає у використанні AMPL (A Mathematical Programming Language) [5] та сучасного програмного забезпечення для розв'язання задач цілочислового лінійного програмування, наприклад, програми gurobi [6].

AMPL-код для опису оптимізаційної задачі (1) – (7) має наступний вигляд:

```
param m>=2; #Кількість постачальників (пункти A)
param l>=1; #Кількість проміжних пунктів (пункти D)
param n>=2; #Кількість споживачів (пункти B)
param d>=1<=m; #границя зверху на Кількість пунктів D
#Вартості перевезення одиниці продукції:
param cik{i in 1..m, k in 1..l} >= 0; #від A до D
param ckj{k in 1..l, j in 1..n} >= 0; #від D до B
#Інші дані
param a{i in 1..m} >= 0; #Продукція в A
param b{j in 1..n} >= 0; #Потреби в B
check: sum{i in 1..m} a[i] = sum{j in 1..n} b[j]; # теорема 1
#Невідомі (продукція, яку потрібно перевезти)
var x{i in 1..m, k in 1..l} >=0; #від A до D
var y{k in 1..l, j in 1..n} >=0; #від D до B
var z{k in 1..l} binary;        # 1-задіяно D, 0-незадіяно D
minimize f_opt: #Мінімізувати витрати на перевезення продукції:
sum{i in 1..m, k in 1..l} cik[i,k]*x[i,k]+ #від A до D
sum{k in 1..l, j in 1..n} ckj[k,j]*y[k,j]; #від D до B
subject to #за обмежень
con2 {i in 1..m}: #перевезення продукції з A до D
      sum{k in 1..l}x[i,k] = a[i];
con3 {j in 1..n}: #задоволення потреб B з D
      sum{k in 1..l}y[k,j] = b[j];
con4 {k in 1..l}: #всю продукцію потрібно доставити в B
      sum{i in 1..m}x[i,k]-sum{j in 1..n}y[k,j]=0;
con5 {k in 1..l}: #вкл./викл. проміжний пункт
sum{i in 1..m}x[i,k]<=z[k]*sum{i in 1..m}a[i];
con6:          #вибрати не більше ніж d проміжних пунктів
sum{k in 1..l}z[k]<=d;
```

тут оператор **param** використано для опису розмірів та даних задачі; оператор **var** – для опису змінних задачі, де враховано їх невід'ємність та булевість; оператор **check** – для перевірки сумісності обмежень (2) – (7) згідно теореми 2, яка вимагає, щоб уся продукція постачальників була відправлена споживачам. Якщо цей AMPL-код доповнити необхідними даними для конкретної задачі, то його можна використовувати для розв'язання задачі (1) – (7) за допомогою стандартного програмного забезпечення для розв'язання задач цілочислового лінійного програмування. Далі наведемо його застосування для програми gurobi та наступного тестового прикладу.

Тестовий приклад. Виробниче об'єднання складається з трьох філіалів: A_1, A_2, A_3, які виготовляють однорідну продукцію в обсягах відповідно 200, 220 та 380 одиниць на місяць. Ця продукція відправляється на три склади D_1, D_2 і D_3, а потім – до чотирьох споживачів B_1, B_2, B_3, B_4, попит яких становить відповідно 150, 50, 350 і 250 одиниць. Вартості перевезень одиниці продукції (в умовних одиницях) від виробників на склади, а потім – із складів до споживачів наведені в табл. 1, 2.

ТАБЛИЦЯ 1

$A \setminus D$	D_1	D_2	D_3
A_1	5	2	5
A_2	3	1	3
A_3	4	7	1

ТАБЛИЦЯ 2

$D \setminus B$	B_1	B_2	B_3	B_4
D_1	3	2	5	2
D_2	7	1	3	1
D_3	8	5	6	5

Для цього прикладу опис вхідних даних задачі (1) – (7) реалізує AMPL-код:

```
data; #Блок вхідних даних, де задаємо:
param m := 3; #Кількість постачальників A
param l := 3; #Кількість проміжних пунктів D
param n := 4; #Кількість споживачів B
param d := 3; #границя зверху на кількість D
#Витрати на перевезення одиниці продукції:
param cik: 1 2 3 := #від A ( ) до D ( )
         1 5 2 5
         2 3 1 3
         3 4 7 1;
param ckj: 1 2 3 4 := #від D ( ) до B ( )
         1 3 2 5 2
         2 7 1 3 1
         3 8 5 6 5;
param a:= #Продукція в A
1 200 2 220 3 380; #A_1, A_2, A_3
param b:= #Потреби B
1 150 2 50 3 350 4 250; #B_1, B_2, B_3, B_4
options solver gurobi;
solve; #Розв'язати задачу (1)-(5)
#Роздрукувати значення цільової функції та час розв'язання
display f_opt, _solve_time;
#Роздрукувати значення оптимальних змінних
display x; display y; display z.
```

Розв'язок задачі (1) – (7) для тестового прикладу при $d = 3$ показаний на рис. 1. Для нього значення цільової функції дорівнює 3940 умовних одиниць, склад D_1 завантажений продукцією на 150 одиниць, склад D_2 – на 420 одиниць, а склад D_3 – на 230 одиниць.



РИС. 1. Оптимальний план перевезень продукції при $d = 3$

Розв'язок задачі (1) – (7) для тестового прикладу при $d = 2$ показаний на рис. 2. Для нього значення цільової функції дорівнює 4170 умовних одиниць, склад D_1 завантажений продукцією на 380 одиниць, склад D_2 – на 420 одиниць, а склад D_3 – не використовується.



РИС. 2. Оптимальний план перевезень продукції при $d = 2$

Висновки. В роботі наведено результати дослідження двоетапної транспортної задачі для визначення найекономічнішого плану перевезення однорідної продукції від постачальників до споживачів, коли кількість проміжних пунктів (сховищ, складів) є обмеженою зверху. Якщо в задачі задіяні всі проміжні пункти, то вона збігається з класичною двоетапною транспортною задачею. Наведена математична модель, що сформульована як задача булевого лінійного програмування. Визначено умови, при яких задача має розв'язок. Розроблено AMPL-код для її розв'язання сучасними солверами лінійного цілочислового програмування та наведено демонстраційний приклад з результатами роботи програми gurobi.

Розроблений AMPL-код планується використовувати при вивченні навчальної дисципліни «Мережеві задачі оптимізації» магістрантами спеціальностей 8.122 «Комп'ютерні науки» та 8.121 «Інженерія програмного забезпечення» ДВНЗ «Ужгородський національний університет» [7]. Модифікації задачі (1) – (7) та двоетапної транспортної задачі (8) – (12) можуть бути використані студентами вищих навчальних закладів для розробки власних програмних проектів, пов'язаних з теорією та методами оптимізації. Так, наприклад, якщо у двоетапній транспортній задачі обмеження (12) замінити на обмеження, які враховують нижні та верхні межі на кількість продукції, що перевозиться, то описаний метод не зміниться, а спеціалізовані алгоритми будуть вимагати значної корекції.

Задача (1) – (7) та її AMPL-код будуть використані при розробці програмного продукту, що зможе обчислювати оптимальний план перевезень однорідної продукції використовуючи гео-дані. Витрати на перевезення продукції між постачальниками та проміжними пунктами, проміжними пунктами та споживачами будуть враховувати відстані між постачальниками, проміжними пунктами і споживачами.

П.И. Стецюк, О.П. Бисага, С.С. Трегубенко

ДВУХЭТАПНАЯ ТРАНСПОРТНАЯ ЗАДАЧА С ОГРАНИЧЕНИЕМ НА КОЛИЧЕСТВО ПРОМЕЖУТОЧНЫХ ПУНКТОВ

Предложена математическая модель двухэтапной транспортной задачи для определения оптимального плана перевозки однородной продукции от поставщиков к потребителям, если количество промежуточных пунктов ограничено сверху. Математическая модель сформулирована как задача булевого линейного программирования. Определены условия, при которых задача имеет решение, и приведен AMPL-код для ее решения современными солверами линейного целочисленного программирования. Приведен демонстрационный пример результатов расчета с помощью программы gurobi.

P.I. Stetsyuk, O.P. Bysaha, S.S. Tregubenko

TWO-STAGE TRANSPORTATION PROBLEM WITH CONSTRAINT ON THE NUMBER OF INTERMEDIATE LOCATIONS

A mathematical model of the two-stage transportation problem is proposed to determine the optimal plan for transportation of homogeneous products from suppliers to consumers if the number of intermediate locations is bounded above. The mathematical model is formulated as a Boolean linear programming problem. The conditions under which the problem has a solution are determined, and AMPL-code for solving the problem by state-of-the-art linear integer programming solvers is given. A demo example of calculation results using *gurobi* program is presented.

Список літератури

1. Михалевич В.С., Трубин В.А., Шор Н.З. Оптимизационные задачи производственно-транспортного планирования. М.: Наука, 1986. 264 с.
2. Карагодова О.О., Кігель В.Р., Рожок В.Д. Дослідження операцій: Навч. посіб. К.: Центр учбової літератури, 2007. 256 с.
3. Наконечний С.І., Савіна С.С. Математичне програмування: Навч. посіб. К.: КНЕУ, 2003. 452 с.
4. Стецюк П.І., Ляшко В.І., Мазютинець Г.В. Двоетапна транспортна задача та її AMPL-реалізація. Наукові записки НаУКМА. *Комп'ютерні науки*. 2018. Т. 1. С. 14–20.
5. Fourer R., Gay D., Kernighan B. AMPL A Modeling Language for Mathematical Programming. Belmont: Duxburry Press, 2003. 517 p.
6. Gurobi Optimization, Inc., Gurobi Optimizer Reference Manual. 2014. http://www.gurobi.com.
7. Стецюк П.І., Міца О.В., Пецко В.І. Мережні інформаційні технології: методичні рекомендації до вивчення курсу. Ужгород: Видавництво УжНУ «Говерла», 2014. 65 с.

Одержано 01.11.2018

Про авторів:

Стецюк Петро Іванович,
доктор фізико-математичних наук,
завідувач відділу Інституту кібернетики імені В.М. Глушкова НАН України,
E-mail: stetsyukp@gmail.com

Бисага Олег Петрович,
аспірант Ужгородського національного університету,
E-mail: olehbusaha@gmail.com

Трегубенко Станіслав Семенович,
доктор військових наук, старший науковий співробітник
Центрального науково-дослідного інституту Збройних Сил України.
E-mail: info-cvni@ukr.net